
*
*              PM Signal
*
*     Vtime       ^  OUT     ^  REF
*        ____    /_\        /_\
*  VT  _|_   |   _|_        _|_
*     /_  \  |  /_  \ B1   /_  \ Vref
*    // \  \ |_// \  \    // \  \
*    \   \//   \   \//    \   \//
*     \___/     \___/      \___/ Reference
*       |         |          |   Signal
*      _|_       _|_        _|_
*      ///       ///        ///
*
*
*   spec  strt_f stop_f step_f vector [vector ...]
*   spec  2      2k     2      v(out)        spec 2Hz->2kHz @ 2Hz steps
*  .tran  TSTEP  TSTOP  TSTART TMAX   ?UIC?
.OPTIONS  GMIN=1e-12    METHOD=trap   ABSTOL=1e-12  TEMP=27 srcsteps = 1  gminsteps = 1 
*======== ====== ====== ====== ====== ====== ====== ====== ====== ====== ======
VT        Vtime  0      PWL    ( 0 0 1 1 )
B1        OUT    0      V =    sin(6.283185307179*200*v(Vtime) +1*sin(6.283185307179*20*v(Vtime)))
Vref      REF    0      dc =   0  sin( 0 1 200 -1.25m)
.tran     500u   1      0      500u  

.control
set       pensize = 1
run
plot      v(out) v(ref) xlimit 0 100m
set       pensize = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
*hanning,cosine,bartlet,blackman,rectangular,hamming,triangle,gaussian
linearize
set       specwindow = "blackman"
spec      1       500   1     v(out)
plot      db(v(out))  
.endc
.end

The total sample period is 1sec(1Hz)
The Carrier is at 200Hz.
Modulation is at 20Hz at 1 radian peak.

B1        OUT    0      V =    sin(6.283185307179*200*v(Vtime) +1*sin(6.283185307179*20*v(Vtime)))
Vref      REF    0      dc =   0  sin( 0 1 200 -1.25m)



Plot both the PM signal and reference signal.


